
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc79761026]Corrections required
0.1 [bookmark: _Toc79761027]Correction to function fl_TC_7_1_2_3_5and5a
	Function name
	fl_TC_7_1_2_3_5and5a

	Reason for change
	In the current TTCN implementation of 7.1.2.3.5 and 7.1.2.3.5a
1) Step 1 is transmitting AMD PDU ((SN = j*iteration_ size, SN=(((j+1)*iteration_size)-1) ) and Step 1A is transmitting one extra RLC PDU ( SN = W ) , it is proposed to merge these two steps , so that it can be done within the same loop structure
2) At step 1 , before sending the RLC PDUs , v_TimingDL is added with (20*p_innerLoop) ms duration which allows UL grant to start much earlier than DL data transmission 

	Summary of change
	1) It is proposed to merge step 1 and step 1A so that TTCN will generate and send RLC PDU (SN = j*iteration_ size, SN=(((j+1)*iteration_size)) in the same loop structure
2) To be in align with Prose , v_TimingDL update before sending the RLC PDUs at step1 has to be removed , so that Step 2 ( UL Grant )  can be started 100ms after the first DL transmission happened in Step 1( as mentioned in the Prose )

	TTCN module
	ttcn\develop\NR_TC_Common\7_1_2\RLC_TC_Common_NR.ttcn

	MCC160 Comment
	



Before Change:
	function fl_TC_7_1_2_3_5and5a(NR_RLC_SS_State_Type p_RLC_Rec,
                                DRB_Identity p_NR_DRB_Id,
                                integer p_OuterLoop,
                                integer p_InnerLoop) runs on NR_BASE_PTC
  { // @sic R5-203535 sic@
    var SubFrameTiming_Type v_TimingDL;
    var SubFrameTiming_Type v_TimingUL;
    var integer v_RLC_SDUsize := 10;
    var integer v_NR_AM_Window_Size;
    var B1_Type v_Poll;
    var integer i;
    var integer j;
    var integer v_SlotsNumber;
    var template (value) NR_DRB_DataPerSlotList_DL_Type v_DataPerSlotList;
    var template (value) NR_DRB_DataPerSlotList_DL_Type v_DataPerSlotListRx;
    var template (value) NR_DRB_DataPerSlotList_DL_Array v_DataPerSlotArray;
    var integer v_VerdictFilter;
    var template (value) NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU_TX;
    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;
    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);
    var NR_RLC_DataFieldList_Type v_NR_RLC_DataList;
    var template (present) NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU_RX;
    var template (present) NR_RLC_AM_StatusPDU_Type v_NR_RLC_StatusPDU;
    var NR_RLC_DataField_Type v_NR_RLC_Data;
    var template (value) NR_RLC_AMD_Data_Type v_NR_RLC_AMD_Data;
    timer t_WatchDog;
    
    if (p_RLC_Rec.AM_SN_Size == 18) {
      v_NR_AM_Window_Size := tsc_NR_AM_Window_Size_18;
      v_RLC_SDUsize := 4; // @sic R5s190730, R5s190772, R5-197011, R5-203535 sic@
      // TBS = RLC PDU data + RLC header without SO (3 octets)+ MAC header (2 octets) + MAC BSR or padding (2 octets) = 4 + 3 + 2 + 2 = 11 octets = 88 bits => TBS = 88 bits
      v_VerdictFilter := 4096; // @sic R5-187106 sic@
    } else { //SN12
      v_NR_AM_Window_Size := tsc_NR_AM_Window_Size_12;
      v_RLC_SDUsize := 4; // @sic R5-185083, R5s190730, R5s190772, R5-197011, R5s190906, R5-203535 sic@
      // TBS = RLC PDU data + RLC header without SO (2 octets)+ MAC header (2 octets) + MAC BSR or padding (2 octets)= 4 + 2 + 2 + 2 = 10 octets = 80 bits => TBS = 88 bits
      v_VerdictFilter := 256; // @sic R5-187106 sic@
    }

    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(88, v_NR_UplinkBWP);
    v_SlotsNumber := f_NR_CellInfo_GetSlotsNumberPerSubframe(nr_Cell1);

    // @siclog "Step 0" siclog@
    // The SS does not allocate any uplink grant.
    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);
    
    // generate W RLC SDUs in lists of  length iteration size
    // p_OuterLoop * p_InnerLoop = W
    for (j := 0; j < p_OuterLoop; j:= j + 1) {
      // generate p_InnerLoop RLC SDUs = PDCP PDUs for transmission
      v_NR_RLC_DataList := fl_NR_GenerateRLC_AM_SDUs(p_RLC_Rec, v_RLC_SDUsize, p_InnerLoop);
      // put those p_InnerLoop AMD PDUs into a list for transmission
      v_Poll := tsc_NR_P_NoPoll; // no indication of Poll bit usage in the prose
      for (i := 0; i < p_InnerLoop; i:= i + 1) {
        v_NR_RLC_AMD_PDU_TX := f_NR_RLC_Get_AMD_FullSDU_TX(p_RLC_Rec, v_Poll, v_NR_RLC_DataList[i]);
        v_DataPerSlotList[i] := cs_NR_DRB_DataPerSlotList_DL_SingleAmdPdu(v_SlotsNumber * 20 * i, -, v_NR_RLC_AMD_PDU_TX);
      }
      v_DataPerSlotArray [j] := v_DataPerSlotList;
    }
    // generate 1 more RLC SDU for transmission
    v_NR_RLC_DataList := fl_NR_GenerateRLC_AM_SDUs(p_RLC_Rec, v_RLC_SDUsize, 1);
    v_NR_RLC_AMD_PDU_TX := f_NR_RLC_Get_AMD_FullSDU_TX(p_RLC_Rec, v_Poll, v_NR_RLC_DataList[0]);
    v_DataPerSlotList[0] := cs_NR_DRB_DataPerSlotList_DL_SingleAmdPdu(0, -, v_NR_RLC_AMD_PDU_TX); // @sic R5s210093 sic@

    // EXCEPTION: The SS is configured for step 1 500 ms in advance. The transmissions are performed every second radio frame.
    // Step 2 is started 100 ms after the first DL AMD PDU has been transmitted in step 1.
    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 500);
    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL, 100);
    
    // @siclog "Step 2" siclog@
    // In the following steps the SS transmits 1 UL grant in every second radio frame to enable the UE to return each received AMD PDU in one looped back AMD PDU.
    // Step 2 executed before step 1 to ensure the UL grant is configured in time
    f_NR_ULGrantConfiguration_StartPeriodicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20);

    // EXCEPTION: Step 1 shall be repeated from j=0 to j=FLOOR((Maximum_RLC_SN/iteration size).
    // @siclog "Step 1" siclog@
    // The SS transmits several RLC PDUs in a RLC PDU List, the number of RLC PDUs sent is defined by the iteration_size.
    // Each RLC Data PDU contains one RLC SDU. The SS transmits an AMD PDU containing a SDU to the UE.
    for (j := 0; j < p_OuterLoop; j:= j + 1) {
      v_TimingDL := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20*p_InnerLoop);
      DRB.send (cas_NR_DRB_COMMON_REQ_DataPerSlot(nr_Cell1,
                                                  p_NR_DRB_Id,
                                                  cs_TimingInfo_NR(v_TimingDL),
                                                  v_DataPerSlotArray [j]));
      v_TimingDL := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20*p_InnerLoop); // @sic R5s210093 sic@
    }

    // @siclog "Step 1A" siclog@
    // The SS transmits one RLC PDU in the slot following the transmissions of step 1.
    DRB.send (cas_NR_DRB_COMMON_REQ_DataPerSlot(nr_Cell1,
                                                p_NR_DRB_Id,
                                                cs_TimingInfo_NR(v_TimingDL),
                                                {v_DataPerSlotList[0]})); // @sic R5s210093 sic@

    // EXCEPTION: Step 2A  shall be repeated from j=0 to j=FLOOR((Maximum_RLC_SN/iteration size).
    // EXCEPTION: In Step 2A, SS shall receive a RLC PDU and step 2A is repeated from SN=j*iteration_size to SN=(((j+1)*iteration_ size)-1).
    // @siclog "Step 2A" siclog@
    // The SS transmits one RLC PDU in the slot following the transmissions of step 1.
    for (j := 0; j < p_OuterLoop; j:= j + 1) {
      v_DataPerSlotListRx := v_DataPerSlotArray [j];
      for (i := 0; i < p_InnerLoop; i:= i + 1) {
      if (p_RLC_Rec.AM_SN_Size == 18) {
        v_NR_RLC_AMD_Data := v_DataPerSlotListRx[i].PduSduList.RlcPdu[0].AMD.SN18Bit.Data; // @sic R5s201490 sic@
      } else {
        v_NR_RLC_AMD_Data := v_DataPerSlotListRx[i].PduSduList.RlcPdu[0].AMD.SN12Bit.Data; // @sic R5s201490 sic@
      }

      if ((j == p_OuterLoop-1) and (i == p_InnerLoop-1)) {// SN = W-1 @sic R5s201490 sic@
        // @siclog "Step 3" siclog@
        // Check: Does the UE transmit the Wth AMD PDU with the Poll bit set and with the contents of the SDU?.
        f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, v_NR_RLC_AMD_Data, tsc_NR_P_Poll);
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3");
      } else {// SN < W-1
        f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, v_NR_RLC_AMD_Data, tsc_NR_P_NoPoll);
      }

      if ((((j*p_OuterLoop) + (i+1)) mod v_VerdictFilter) == 0) {f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2A" & " SN = " &  int2str(((j*p_OuterLoop) + i)))};
      }
    }
   
    // @siclog "Step 4" siclog@
    // The SS starts the UL default grant transmission.
    f_NR_ULGrantConfiguration_Start(nr_Cell1);
    
    // @siclog "Step 5" siclog@
    // Check: Does the UE transmit an AMD PDU within t-PollRetransmit/2?
    t_WatchDog.start(p_RLC_Rec.t_PollRetransmit/2.0);
    alt {
      [] DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1, p_NR_DRB_Id))
        {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 5");
        }
      [] t_WatchDog.timeout
        {}
    }
    
    // @siclog "Step 6" siclog@
    // The SS transmits a STATUS PDU to acknowledge the W uplink AMD PDUs with SN=0 to SN=W-1. ACK_SN = W.
    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id);
    
    // @siclog "Step 7" siclog@
    // Check: Does the UE transmit the (w+1)st AMD PDU with the Poll bit set and with the contents of the SDU?
    f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, v_NR_RLC_DataList[0], tsc_NR_P_Poll);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 7");
    
    // @siclog "Step 8" siclog@
    // The SS transmits a STATUS PDU with ACK_SN = W+1.
    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Step 9" siclog@
    // The SS transmits the (W+2)nd AMD PDU containing a SDU to the UE with the Sequence Number field set to ((2W+1 mod AM_Modulus) = 1) and the Polling bit set.
    // generate data for 1 more RLC SDU (w+2nd)
    v_NR_RLC_DataList := fl_NR_GenerateRLC_AM_SDUs(p_RLC_Rec, v_RLC_SDUsize, 1);
    p_RLC_Rec.AM_TX_Next := 1; // @sic R5s190730, R5s190772, R5-197011 sic@
    v_NR_RLC_Data := f_NR_PDCP_PDU_Encvalue(cs_NR_PDCP_PDU_18B((v_NR_AM_Window_Size + 1), substr(v_NR_RLC_DataList[0], 3, lengthof (v_NR_RLC_DataList[0]) - 3)));
    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, v_NR_RLC_Data, tsc_NR_P_Poll);
    
    // @siclog "Step 10" siclog@
    // Check: Does the UE transmit a STATUS PDU acknowledging W+1 SDUs? (ACK_SN = W+1).
    p_RLC_Rec.AM_RX_Next_Ack := v_NR_AM_Window_Size + 1;
    f_NR_RxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10");
    
    // @siclog "Step 11" siclog@
    // The SS transmits the (W+2)nd AMD PDU containing a SDU to the UE with the Sequence Number field set to W+1 and the Polling bit set.
    p_RLC_Rec.AM_TX_Next := v_NR_AM_Window_Size + 1; // @sic R5s190730, R5s190772 sic@
    v_NR_RLC_Data := f_NR_PDCP_PDU_Encvalue(cs_NR_PDCP_PDU_18B((v_NR_AM_Window_Size + 1), substr(v_NR_RLC_DataList[0], 3, lengthof (v_NR_RLC_DataList[0]) - 3)));
    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, v_NR_RLC_Data, tsc_NR_P_Poll);

    // EXCEPTION: Step 12 and 13 can happen in any order @sic R5s200148, R5s200164 sic@
    // @siclog "Step 12" siclog@
    p_RLC_Rec.AM_RX_Next_Ack := v_NR_AM_Window_Size + 2; // @sic R5s190730, R5s190772 sic@
    // Check: Does the UE transmit a STATUS PDU acknowledging W+1 PDUs? (ACK_SN field = W+2)?
    // @siclog "Step 13" siclog@
    // Check: Does the UE transmit an AMD PDU with the same data as received in the corresponding DL AMD PDU in step 11?.
    v_NR_RLC_AMD_PDU_RX := f_NR_RLC_Get_AMD_FullSDU_RX(p_RLC_Rec, tsc_NR_P_Poll, v_NR_RLC_Data);
    v_NR_RLC_StatusPDU := f_NR_RLC_Get_STATUS_PDU_RX(p_RLC_Rec);
    
      alt { //Expect RLC AMD PDCU and Status PDU in same or in multiple slots but in any order
      [] DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,
                                                        p_NR_DRB_Id,
                                                        -,
                                                        {cr_NR_RLC_PDU_Status(v_NR_RLC_StatusPDU)} )) 
      {
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12");
        DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,
                                                        p_NR_DRB_Id,
                                                        -,
                                                        {cr_NR_RLC_PDU_AMD(v_NR_RLC_AMD_PDU_RX)} )); 
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 13");
      }
      [] DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,
                                                        p_NR_DRB_Id,
                                                        -,
                                                        {cr_NR_RLC_PDU_AMD(v_NR_RLC_AMD_PDU_RX)} )) {
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 13");
        DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,
                                                        p_NR_DRB_Id,
                                                        -,
                                                        {cr_NR_RLC_PDU_Status(v_NR_RLC_StatusPDU)} ));
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12");
      }
      [] DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,
                                                        p_NR_DRB_Id,
                                                        -,
                                                        { cr_NR_RLC_PDU_Status(v_NR_RLC_StatusPDU), cr_NR_RLC_PDU_AMD(v_NR_RLC_AMD_PDU_RX)})) {
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12 and 13");
      }
      [] DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,
                                                        p_NR_DRB_Id,
                                                        -,
                                                        { cr_NR_RLC_PDU_AMD(v_NR_RLC_AMD_PDU_RX), cr_NR_RLC_PDU_Status(v_NR_RLC_StatusPDU)})) {
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12 and 13");
      }
    }  

    // @siclog "Step 14" siclog@
    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id); // @sic R5s190730, R5s190772, R5-197011 sic@
  }



After Change:
	  function fl_TC_7_1_2_3_5and5a(NR_RLC_SS_State_Type p_RLC_Rec,
                                DRB_Identity p_NR_DRB_Id,
                                integer p_OuterLoop,
                                integer p_InnerLoop) runs on NR_BASE_PTC
  { // @sic R5-203535 sic@
    var SubFrameTiming_Type v_TimingDL;
    var SubFrameTiming_Type v_TimingUL;
    var integer v_RLC_SDUsize := 10;
    var integer v_NR_AM_Window_Size;
    var B1_Type v_Poll;
    var integer i;
    var integer j;
    var integer v_SlotsNumber;
    var template (value) NR_DRB_DataPerSlotList_DL_Type v_DataPerSlotList;
    var template (value) NR_DRB_DataPerSlotList_DL_Type v_DataPerSlotListRx;
    var template (value) NR_DRB_DataPerSlotList_DL_Array v_DataPerSlotArray;
    var integer v_VerdictFilter;
    var template (value) NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU_TX;
    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;
    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);
    var NR_RLC_DataFieldList_Type v_NR_RLC_DataList;
    var template (present) NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU_RX;
    var template (present) NR_RLC_AM_StatusPDU_Type v_NR_RLC_StatusPDU;
    var NR_RLC_DataField_Type v_NR_RLC_Data;
    var template (value) NR_RLC_AMD_Data_Type v_NR_RLC_AMD_Data;
    timer t_WatchDog;
    
    if (p_RLC_Rec.AM_SN_Size == 18) {
      v_NR_AM_Window_Size := tsc_NR_AM_Window_Size_18;
      v_RLC_SDUsize := 4; // @sic R5s190730, R5s190772, R5-197011, R5-203535 sic@
      // TBS = RLC PDU data + RLC header without SO (3 octets)+ MAC header (2 octets) + MAC BSR or padding (2 octets) = 4 + 3 + 2 + 2 = 11 octets = 88 bits => TBS = 88 bits
      v_VerdictFilter := 4096; // @sic R5-187106 sic@
    } else { //SN12
      v_NR_AM_Window_Size := tsc_NR_AM_Window_Size_12;
      v_RLC_SDUsize := 4; // @sic R5-185083, R5s190730, R5s190772, R5-197011, R5s190906, R5-203535 sic@
      // TBS = RLC PDU data + RLC header without SO (2 octets)+ MAC header (2 octets) + MAC BSR or padding (2 octets)= 4 + 2 + 2 + 2 = 10 octets = 80 bits => TBS = 88 bits
      v_VerdictFilter := 256; // @sic R5-187106 sic@
    }

    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(88, v_NR_UplinkBWP);
    v_SlotsNumber := f_NR_CellInfo_GetSlotsNumberPerSubframe(nr_Cell1);

    // @siclog "Step 0" siclog@
    // The SS does not allocate any uplink grant.
    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);
    
    // generate W RLC SDUs in lists of  length iteration size
    // p_OuterLoop * p_InnerLoop = W
    for (j := 0; j < p_OuterLoop; j:= j + 1) {
      // generate p_InnerLoop RLC SDUs = PDCP PDUs for transmission
      // In case of 7.1.2.3.5 if block will be executed to generate 1 extra RLC packet + 2048 packets
      // In case of 7.1.2.3.5a if block will be executed to generate 1 extra RLC packet + 2048 packets for the last outerloop iteration
      if ( p_OuterLoop - j == 1) 
      { 
          v_NR_RLC_DataList := fl_NR_GenerateRLC_AM_SDUs(p_RLC_Rec, v_RLC_SDUsize, p_InnerLoop + 1); 
          // put those p_InnerLoop AMD PDUs into a list for transmission
          v_Poll := tsc_NR_P_NoPoll; // no indication of Poll bit usage in the prose
          for (i := 0; i < (p_InnerLoop + 1); i:= i + 1) { 
          v_NR_RLC_AMD_PDU_TX := f_NR_RLC_Get_AMD_FullSDU_TX(p_RLC_Rec, v_Poll, v_NR_RLC_DataList[i]); 
          v_DataPerSlotList[i] := cs_NR_DRB_DataPerSlotList_DL_SingleAmdPdu(v_SlotsNumber * 20 * i, -, v_NR_RLC_AMD_PDU_TX); 
          } 
          v_DataPerSlotArray [j] := v_DataPerSlotList;
      } 
      // In case of 7.1.2.3.5a else block will be executed to generate 2048 packets for the (Outerloop-1) iterations
      else 
      { 
          v_NR_RLC_DataList := fl_NR_GenerateRLC_AM_SDUs(p_RLC_Rec, v_RLC_SDUsize, p_InnerLoop); 
          // put those p_InnerLoop AMD PDUs into a list for transmission 
          v_Poll := tsc_NR_P_NoPoll; // no indication of Poll bit usage in the prose 
          for (i := 0; i < (p_InnerLoop); i:= i + 1) { 
        v_NR_RLC_AMD_PDU_TX := f_NR_RLC_Get_AMD_FullSDU_TX(p_RLC_Rec, v_Poll, v_NR_RLC_DataList[i]);
        v_DataPerSlotList[i] := cs_NR_DRB_DataPerSlotList_DL_SingleAmdPdu(v_SlotsNumber * 20 * i, -, v_NR_RLC_AMD_PDU_TX);
      }
      v_DataPerSlotArray [j] := v_DataPerSlotList;
      } 
    }
    /* WA#7_1_2_3_5 REMOVED generate 1 more RLC SDU for transmission //To be removed and merged in above loop
    v_NR_RLC_DataList := fl_NR_GenerateRLC_AM_SDUs(p_RLC_Rec, v_RLC_SDUsize, 1); 
    v_NR_RLC_AMD_PDU_TX := f_NR_RLC_Get_AMD_FullSDU_TX(p_RLC_Rec, v_Poll, v_NR_RLC_DataList[0]); 
    v_DataPerSlotList[0] := cs_NR_DRB_DataPerSlotList_DL_SingleAmdPdu(v_SlotsNumber * 20 * i, -, v_NR_RLC_AMD_PDU_TX);*/ 

    // EXCEPTION: The SS is configured for step 1 500 ms in advance. The transmissions are performed every second radio frame.
    // Step 2 is started 100 ms after the first DL AMD PDU has been transmitted in step 1.

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 500);
    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL, 100);
    
    // @siclog "Step 2" siclog@
    // In the following steps the SS transmits 1 UL grant in every second radio frame to enable the UE to return each received AMD PDU in one looped back AMD PDU.
    // Step 2 executed before step 1 to ensure the UL grant is configured in time
    f_NR_ULGrantConfiguration_StartPeriodicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20);

    // EXCEPTION: Step 1 shall be repeated from j=0 to j=FLOOR((Maximum_RLC_SN/iteration size).
    // @siclog "Step 1" siclog@
    // The SS transmits several RLC PDUs in a RLC PDU List, the number of RLC PDUs sent is defined by the iteration_size.
    // Each RLC Data PDU contains one RLC SDU. The SS transmits an AMD PDU containing a SDU to the UE.
    for (j := 0; j < p_OuterLoop; j:= j + 1) {
      //v_TimingDL := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20*p_InnerLoop); //To be removed as it makes DL RLC PDU Tx and UL Grant transmission not separated by 100ms
      DRB.send (cas_NR_DRB_COMMON_REQ_DataPerSlot(nr_Cell1,
                                                  p_NR_DRB_Id,
                                                  cs_TimingInfo_NR(v_TimingDL),
                                                  v_DataPerSlotArray [j]));
      v_TimingDL := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20*p_InnerLoop); // @sic R5s210093 sic@
    }

    // @siclog "Step 1A" siclog@
    // The SS transmits one RLC PDU in the slot following the transmissions of step 1. //To be removed as handled in above loop
    /*DRB.send (cas_NR_DRB_COMMON_REQ_DataPerSlot(nr_Cell1, 
                                                p_NR_DRB_Id, 
                                                cs_TimingInfo_NR(v_TimingDL), 
                                                v_DataPerSlotList));*/

    // EXCEPTION: Step 2A  shall be repeated from j=0 to j=FLOOR((Maximum_RLC_SN/iteration size).
    // EXCEPTION: In Step 2A, SS shall receive a RLC PDU and step 2A is repeated from SN=j*iteration_size to SN=(((j+1)*iteration_ size)-1).
    // @siclog "Step 2A" siclog@
    // The SS transmits one RLC PDU in the slot following the transmissions of step 1.
    for (j := 0; j < p_OuterLoop; j:= j + 1) {
      v_DataPerSlotListRx := v_DataPerSlotArray [j];
      for (i := 0; i < p_InnerLoop; i:= i + 1) {
      if (p_RLC_Rec.AM_SN_Size == 18) {
        v_NR_RLC_AMD_Data := v_DataPerSlotListRx[i].PduSduList.RlcPdu[0].AMD.SN18Bit.Data; // @sic R5s201490 sic@
      } else {
        v_NR_RLC_AMD_Data := v_DataPerSlotListRx[i].PduSduList.RlcPdu[0].AMD.SN12Bit.Data; // @sic R5s201490 sic@
      }

      if ((j == p_OuterLoop-1) and (i == p_InnerLoop-1)) {// SN = W-1 @sic R5s201490 sic@
        // @siclog "Step 3" siclog@
        // Check: Does the UE transmit the Wth AMD PDU with the Poll bit set and with the contents of the SDU?.
        f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, v_NR_RLC_AMD_Data, tsc_NR_P_Poll);
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3");
      } else {// SN < W-1
        f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, v_NR_RLC_AMD_Data, tsc_NR_P_NoPoll);
      }

      if ((((j*p_OuterLoop) + (i+1)) mod v_VerdictFilter) == 0) {f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2A" & " SN = " &  int2str(((j*p_OuterLoop) + i)))};
      }
    }
   
    // @siclog "Step 4" siclog@
    // The SS starts the UL default grant transmission.
    f_NR_ULGrantConfiguration_Start(nr_Cell1);
    
    // @siclog "Step 5" siclog@
    // Check: Does the UE transmit an AMD PDU within t-PollRetransmit/2?
    t_WatchDog.start(p_RLC_Rec.t_PollRetransmit/2.0);
    alt {
      [] DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1, p_NR_DRB_Id))
        {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 5");
        }
      [] t_WatchDog.timeout
        {}
    }
    
    // @siclog "Step 6" siclog@
    // The SS transmits a STATUS PDU to acknowledge the W uplink AMD PDUs with SN=0 to SN=W-1. ACK_SN = W.
    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id);
    
    // @siclog "Step 7" siclog@
    // Check: Does the UE transmit the (w+1)st AMD PDU with the Poll bit set and with the contents of the SDU?
    f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, v_NR_RLC_DataList[i], tsc_NR_P_Poll); 
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 7");
    
    // @siclog "Step 8" siclog@
    // The SS transmits a STATUS PDU with ACK_SN = W+1.
    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Step 9" siclog@
    // The SS transmits the (W+2)nd AMD PDU containing a SDU to the UE with the Sequence Number field set to ((2W+1 mod AM_Modulus) = 1) and the Polling bit set.
    // generate data for 1 more RLC SDU (w+2nd)
    v_NR_RLC_DataList := fl_NR_GenerateRLC_AM_SDUs(p_RLC_Rec, v_RLC_SDUsize, 1);
    p_RLC_Rec.AM_TX_Next := 1; // @sic R5s190730, R5s190772, R5-197011 sic@
    v_NR_RLC_Data := f_NR_PDCP_PDU_Encvalue(cs_NR_PDCP_PDU_18B((v_NR_AM_Window_Size + 1), substr(v_NR_RLC_DataList[0], 3, lengthof (v_NR_RLC_DataList[0]) - 3)));
    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, v_NR_RLC_Data, tsc_NR_P_Poll);
    
    // @siclog "Step 10" siclog@
    // Check: Does the UE transmit a STATUS PDU acknowledging W+1 SDUs? (ACK_SN = W+1).
    p_RLC_Rec.AM_RX_Next_Ack := v_NR_AM_Window_Size + 1;
    f_NR_RxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10");
    
    // @siclog "Step 11" siclog@
    // The SS transmits the (W+2)nd AMD PDU containing a SDU to the UE with the Sequence Number field set to W+1 and the Polling bit set.
    p_RLC_Rec.AM_TX_Next := v_NR_AM_Window_Size + 1; // @sic R5s190730, R5s190772 sic@
    v_NR_RLC_Data := f_NR_PDCP_PDU_Encvalue(cs_NR_PDCP_PDU_18B((v_NR_AM_Window_Size + 1), substr(v_NR_RLC_DataList[0], 3, lengthof (v_NR_RLC_DataList[0]) - 3)));
    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, v_NR_RLC_Data, tsc_NR_P_Poll);

    // EXCEPTION: Step 12 and 13 can happen in any order @sic R5s200148, R5s200164 sic@
    // @siclog "Step 12" siclog@
    p_RLC_Rec.AM_RX_Next_Ack := v_NR_AM_Window_Size + 2; // @sic R5s190730, R5s190772 sic@
    // Check: Does the UE transmit a STATUS PDU acknowledging W+1 PDUs? (ACK_SN field = W+2)?
    // @siclog "Step 13" siclog@
    // Check: Does the UE transmit an AMD PDU with the same data as received in the corresponding DL AMD PDU in step 11?.
    v_NR_RLC_AMD_PDU_RX := f_NR_RLC_Get_AMD_FullSDU_RX(p_RLC_Rec, tsc_NR_P_Poll, v_NR_RLC_Data);
    v_NR_RLC_StatusPDU := f_NR_RLC_Get_STATUS_PDU_RX(p_RLC_Rec);
    
   
    alt { //Expect RLC AMD PDCU and Status PDU in same or in multiple slots but in any order
      [] DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,
                                                        p_NR_DRB_Id,
                                                        -,
                                                        {cr_NR_RLC_PDU_Status(v_NR_RLC_StatusPDU)} )) 
      {
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12");
        DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,
                                                        p_NR_DRB_Id,
                                                        -,
                                                        {cr_NR_RLC_PDU_AMD(v_NR_RLC_AMD_PDU_RX)} )); 
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 13");
      }
      [] DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,
                                                        p_NR_DRB_Id,
                                                        -,
                                                        {cr_NR_RLC_PDU_AMD(v_NR_RLC_AMD_PDU_RX)} )) {
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 13");
        DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,
                                                        p_NR_DRB_Id,
                                                        -,
                                                        {cr_NR_RLC_PDU_Status(v_NR_RLC_StatusPDU)} ));
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12");
      }
      [] DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,
                                                        p_NR_DRB_Id,
                                                        -,
                                                        { cr_NR_RLC_PDU_Status(v_NR_RLC_StatusPDU), cr_NR_RLC_PDU_AMD(v_NR_RLC_AMD_PDU_RX)})) {
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12 and 13");
      }
      [] DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,
                                                        p_NR_DRB_Id,
                                                        -,
                                                        { cr_NR_RLC_PDU_AMD(v_NR_RLC_AMD_PDU_RX), cr_NR_RLC_PDU_Status(v_NR_RLC_StatusPDU)})) {
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12 and 13");
      }
    }  

    // @siclog "Step 14" siclog@
    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id); // @sic R5s190730, R5s190772, R5-197011 sic@

  }



